

# **Univerzita Hradec Králové**

Fakulta informatiky a managementu

## **System na identifikaci brouků**

2. projekt ze Znalostních technologií 2

**Petr Voborník**

UHK – FIM - im(5) – 3. ročník – ZT2 - cvičení 05

[vobornik@mikmik.cz](mailto:vobornik@mikmik.cz)

# Obsah

Obsah.....	1
Výběr problémové oblasti .....	2
Postup výběru .....	2
Tabulka vlastností pokusných brouků .....	3
Povolené hodnoty .....	5
Barva .....	5
Prvky .....	5
Předek.....	5
Velké obrázky brouků .....	6
Cyclommatus metalifer .....	6
Cyclommatus mnischechi.....	6
Dicronorhina derbyana var. layardi Péringuey.....	7
Dicronorhina micans.....	7
Dorcus curvidens binodulus Waterhouse .....	8
Dynastes tityus Linnaeus.....	8
Eudicella sp. ....	9
Goliathus meleagris Sjöstedt.....	9
Cheirotonus macleay ssp. formosana Ohaus.....	10
Chelorrhina savagei .....	10
Chlorocala africana oertzeni.....	11
Pachnoda interrupta Olivier.....	11
Pachnoda marginata aurantia .....	12
Prosopocoilus astacoides blanchardi .....	12
Stephanorrhina guttata guttata .....	13
Ovládání programu.....	14
Popis zdrojového kódu .....	15
Fakta.....	15
Zadání pozorovaných hodnot.....	15
Průběžné hledání jediného řešení .....	15
Průběžné zjištění neexistence řešení .....	15
Vypsání výsledků hledání .....	15
Dotaz na pokračování .....	15
Závěr .....	16
Zdrojový kód .....	17

# Výběr problémové oblasti









## *Postup výběru*








Při výběru vhodné problémové oblasti jsem nejprve zvolil motýly. Nicméně záhy jsem zjistil, že jejich identifikace není vůbec snadná, vzhledem k faktu, že stejní motýli měli na různých fotografiích různé barvy, nemluvě o tom, že vrchní a spodní strana křídla si vůbec nebyly podobné.

Jako druhou možnost jsem zvolil brouky. Ti jsou barevní pouze z jedné strany a také na zdroji, z něhož jsem čerpal (<http://vivaex.b7.cz/brouci.htm>), měl každý brouk pouze jedinou fotografii a nedocházelo tak k jejich různému vyobrazení. Vybral jsem tedy 15 různých, výrazně vypadajících, brouků, u nichž jsem poté začal hledat charakteristiky, které by je dokázaly rozlišit. Volba barvy (barev) krovek byla jasná. Za další poznávací znamení jsem určil přední zvláštnost brouka – jeho párový orgán u hlavy. Zvolení brouci měli buď tykadla, rohy (pod ně spadají klepeta a kusadla) nebo nic. Poté jsem zkoušel brouky rozlišovat pomocí jejich rozměrů, avšak ty měli vždy takový rozsah, že téměř u všech se dal najít společný průnik, a tak jsem od této charakteristiky upustil. Místo ní jsem přidal rozlišování pomocí prvků vyobrazených jinou barvou na krovkách. Ve vybrané skupině jsem našel puntíky, pruhy, skvrny, ornamenty případně nic (jednobarevnost).

Po stanovení těchto charakteristik, již bylo možné jednoznačně identifikovat každého z brouků ve výběrové skupině. Všechny jsem sepsal do níže uvedené tabulky i s jejich charakteristikami, kde se o tom je možné přesvědčit. Tabulka je seřazena abecedně. Pro lepší představu jsou v další kapitole uvedeny i zvětšené obrázky brouků.

### Tabulka vlastností pokusných brouků

Obrázek	Název	Barva	Prvky	Ve předu
	Cyclommatus metalifer	tmavě červená		rohý
	Cyclommatus mniszehi	hnědá		rohý
	Dicronorhina derbyana var. layardi Péringuey	zelená červená bílá	pruhy	
	Dicronorhina micans	zelená		tykadla
	Dorcus curvidens binodulus Waterhouse	černá		rohý
	Dynastes tityus Linnaeus	žlutá černá	skvrny	rohý
	Eudicella sp.	červená zelená		tykadla
	Goliathus meleagris Sjöstedt	bílá černá	pruhy	

	Cheirotonus macleay ssp. formosana Ohaus	černá žlutá	skvrny	
	Chelorrhina savagei	zelená žlutá	pruhy	tykadla
	Chlorocala africana oertzeni	tmavě modrá		tykadla
	Pachnoda interrupta Olivier	žlutá černá	ornamenty	tykadla
	Pachnoda marginata aurantia	hnědá žlutá		tykadla
	Prosopocoilus astacoides blanchardi	tmavě žlutá		rohy
	Stephanorrhina guttata guttata	zelená	puntíky	

## ***Povolené hodnoty***

Zde jsou uvedeny povolené hodnoty jednotlivých charakteristik. Jsou zde uvedeny všechny hodnoty, vyskytující se ve výběrové skupině.

### **Barva**

- tmavě červená
- hnědá
- černá
- tmavě žlutá
- žlutá
- zelená
- červená
- tmavě modrá
- bílá

### **Prvky**

- skvrny
- ornamenty
- puntíky
- pruhy

### **Předek**

- rohy
- tykadla
- nic

## **Velké obrázky brouků**

### **Cyclommatus metalifer**



### **Cyclommatus mnischechi**



**Dicronorhina derbyana var. layardi Péringuey**



**Dicronorhina micans**





**Dorcus curvidens binodulus Waterhouse**



**Dynastes tityus Linnaeus**



**Eudicella sp.**



**Goliathus meleagris Sjöstedt**



**Cheirotonus macleay ssp. formosana Ohaus**




**Chelorrhina savagei**



**Chlorocala africana oertzeni**



**Pachnoda interrupta Olivier**



**Pachnoda marginata aurantia**



**Prosopocoilus astacoides blanchardi**



**Stephanorrhina guttata guttata**



## Ovládání programu

- Uživatel spustí program (načte jej – File Open, Load Buffer; resetuje Ctrl+E a a spustí Ctrl+R).
- Okamžitě je dotázán na první charakteristiku brouka – jeho barvu (případně barvy). Zadávají se pouze barvy broukových krovek (zad). Je uveden výčet povolených barev. Pokud je brouk více barvený, je možné vypsát všechny jeho barvy oddělené mezerou. Vypsání barev, je třeba potvrdit stiskem klávesy Enter.
- Pokud zadaným barvám odpovídá pouze jediný brouk v databázi, je vypsáno jeho jméno a dotazování ukončeno. Stejně tak, pokud zadaná kombinace barev neodpovídá žádnému z brouků.
- Je-li v databázi více brouků zadaných barev, je uživatel dotázán na další jeho charakteristiku – tvar barevných prvků (vzorků) na jeho krovkách (zádech). Povolený výčet těchto prvků je uveden. Jde o pruhy (delší souvislé čáry), puntíky (pravidelně uspořádaná kolečka), skvrny (nepravidelně uspořádané fleky různých tvarů), ornamenty (složitější nákres osově souměrný podle středu zad) nebo nic (jednobarevná záda). Uvést lze právě jednu z těchto charakteristik.
- Opět pokud ze zadaných charakteristik již lze nalézt pouze jediného brouka, je vypsáno jeho jméno a dotazování tím končí. Stejně tak, pokud zadaná kombinace neodpovídá žádnému z brouků.
- Další z charakteristik, které je třeba zadat je tvar předku brouka – párový orgán u jeho hlavy. Jde buď o rohy (delší pevná kusadla či klepeta), tykadla (tenčí ohebná) nebo nic (pouze hlava nebo nějaký nepárový orgán).
- Nyní již je vypsán výsledek hledání. Může to být buď jeden či více brouků, kteří odpovídají zadaným charakteristikám nebo oznámení, že brouk s těmito vlastnostmi se v databázi nevyskytuje.
- Dále je uživatel dotázán, přeje-li si pokračovat další identifikací (začít od začátku). Zadá-li jedničku (1), program se spustí znovu, zadá-li nulu (0), program se ukončí.

## Popis zdrojového kódu

### **Fakta**

V této sekci jsou nadefinována fakta: šablona pro definici brouka, jejich databáze, pomocné fakty `colors` (s výčtem povolených barev), `elements` (s výčtem povolených barevných prvků) a `fronts` (s výčtem možných předků brouků). Pomocný fakt `xx`, slouží k možnosti načtení výchozí neplatné množiny.

### **Zadání pozorovaných hodnot**

Pravidla v této sekci se postupně dotazují uživatele na jednotlivé charakteristiky pozorovaného brouka. Jsou řazena tak, aby ta dříve zadaná, měla větší výběrovou vlastnost (vyřadila z možného výběru více kandidátů). Veškeré zadávané hodnoty prochází validací správnosti zadání a pokud nevyhovují, uživatel je na příslušnou hodnotu dotázán znovu.

### **Průběžné hledání jediného řešení**

Tato pravidla, s vyšší spouštěcí prioritou, než ta předchozí, se po zadání každé jednotlivé charakteristiky pokouší v databázi zjistit, jestli zadané vlastnosti již nestačí k tomu, aby byl nalezen jediný brouk. Pokud ano, vypíše jej a uživatele se dále na nic nedotazují. Další upřesnění by totiž vedlo buď ke stejnému či nulovému výsledku.

### **Průběžné zjištění neexistence řešení**

Jelikož je možné, že uživatel zadá neexistující kombinaci charakteristik brouka, dříve než je dotázán na všechny možnosti, je zbytečné se jej od této chvíle dále dotazovat. O to se starají pravidla v této sekci. Pokud program zjistí, že zadaným vlastnostem neodpovídá žádný brouk v databázi, informuje o této skutečnosti uživatele a ukončí další dotazování.

### **Vypsání výsledků hledání**

Tato sekce se stará o výpis výsledků hledání, po zadání všech charakteristik. Pravidlo `brouci_jsou` vypíše všechny brouky, které odpovídají zadaným charakteristikám (pokud jich je víc, či jen jeden) a pravidlo `brouci_nejsou` informuje uživatele, že brouk se zadanými vlastnostmi v databázi není.

### **Dotaz na pokračování**

Po vypsání výsledků se pravidla v této sekci uživatele zeptají, přeje-li si pokračovat další identifikací a pokud ano, odstraní pomocné fakty z předchozího hledání, čímž se program znovu spustí od začátku.



## **Závěr**

Zadaný problém se mi podařilo zpracovat úspěšně. Program přesně plní daný účel, je odolný vůči chybným vstupům a uživatelsky přívětiví.

Do systému je možno přidávat i další brouky, které lze popsat zadanými charakteristikami. V tom případě však je možné, že pro jedinečnou identifikaci každého z nich bude třeba přidat ještě některou další vlastnost.

## Zdrojový kód

```
; ----- Fakta -----

(deftemplate brouk
  (slot jmeno (type SYMBOL))
  (slot predek (allowed-symbols rohy tykadla nic))
  (slot prvky (allowed-symbols skvrny ornamenty puntiky pruhu zadne))
  (multislot barva (type SYMBOL)
    (allowed-symbols bila zluta zelena cervena hneda tmave_zluta
tmave_cervena tmave_modra cerna))
)

(deffacts data
  (brouk (jmeno cyclommatus_metalifer) (predek rohy)
  (prvky zadne) (barva tmave_cervena))
  (brouk (jmeno cyclommatus_mniszechi) (predek rohy)
  (prvky zadne) (barva hneda))
  (brouk (jmeno dicronorhina_derbyana_var_layardi_peringuey) (predek nic)
  (prvky pruhu) (barva zelena cervena bila))
  (brouk (jmeno dicronorhina_micans) (predek
tykadla) (prvky zadne) (barva zelena))
  (brouk (jmeno dorcus_curvidens_binodulus_waterhouse) (predek rohy)
  (prvky zadne) (barva cerna))
  (brouk (jmeno dynastes_tityus_linnaeus) (predek rohy)
  (prvky skvrny) (barva zluta cerna))
  (brouk (jmeno eudicella_sp) (predek
tykadla) (prvky zadne) (barva cervena zelena))
  (brouk (jmeno goliathus_meleagris_sjostedt) (predek nic)
  (prvky pruhu) (barva bila cerna))
  (brouk (jmeno cheirotonus_macleay_ssp_formosana_ohaus) (predek nic)
  (prvky skvrny) (barva zluta cerna))
  (brouk (jmeno chelorrhina_savagei) (predek
tykadla) (prvky pruhu) (barva zelena zluta))
  (brouk (jmeno chlorocala_africana_oertzeni) (predek
tykadla) (prvky zadne) (barva tmave_modra))
  (brouk (jmeno pachnoda_interrupta_olivier) (predek
tykadla) (prvky ornamenty) (barva cerna zluta))
  (brouk (jmeno pachnoda_marginata_aurantia) (predek
tykadla) (prvky zadne) (barva hneda zluta))
  (brouk (jmeno prosopocoilus_astacoides_blancharidi) (predek rohy)
  (prvky zadne) (barva tmave_zluta))
  (brouk (jmeno stephanorrhina_guttata_guttata) (predek nic)
  (prvky puntiky) (barva zelena))

  (colors bila zluta zelena cervena hneda tmave_zluta tmave_cervena
tmave_modra cerna)
  (elements skvrny ornamenty puntiky pruhu zadne)
  (fronts rohy tykadla nic)
  (xx xx)
)
```

```

; ----- Zadani pozorovanych hodnot -----

(defrule barvy
  (not (konec ?k))
  (not (barva $?b)) (not (prvek ?p)) (not (predek ?r))
  (colors $?colors)
  (xx $?xx)
  =>
  (bind $?barvy $?xx)
  (while (not(subsetp $?barvy $?colors)) do
    (printout t "Jakou ma brouk barvu/barvy krovek (svrchni casti)? Vypiste
vsechny barvy (oddelene mezerou)." crlf
      "Povolene barvy jsou: " $?colors crlf)
    (bind $?barvy (explode$ (readline)))
  )
  (assert (barva $?barvy))
  (printout t crlf)
)

(defrule prvky
  (not (konec ?k))
  (barva $?b) (not (prvek ?p)) (not (predek ?r))
  (elements $?elements)
  (xx ?xx)
  =>
  (bind ?prvky ?xx)
  (while (not(member$ ?prvky $?elements)) do
    (printout t "Jaka prvky (vzorky) ma brouk na svych krovkach (svrchni
casti)? Uvedte pouze jeden." crlf
      "Zname prvky jsou: " $?elements crlf)
    (bind ?prvky (read))
  )
  (assert (prvek ?prvky))
  (printout t crlf)
)

(defrule predky
  (not (konec ?k))
  (barva $?b) (prvek ?p) (not (predek ?r))
  (fronts $?fronts)
  (xx ?xx)
  =>
  (bind ?predky ?xx)
  (while (not(member$ ?predky $?fronts)) do
    (printout t "Jaka je parova zvlastnost na predni casti brouka (na
hlave)? Uvedte pouze jednu." crlf
      "Zname druhy jsou: " $?fronts crlf)
    (bind ?predky (read))
  )
  (assert (predek ?predky))
  (printout t crlf)
)

```

```

; ----- Prubezne hledani jedineho reseni -----

(defrule jeden_brouk_je_barva
  (declare (auto-focus TRUE) (salience 10))
  (not (konec ?k))
  (barva $?barvy)
  (not (prvek ?p)) (not (predek ?r))
  (brouk (jmeno ?jmeno) (barva $?barvy2&:(subsetp $?barvy $?barvy2)))
  (not(brouk (jmeno ?jmeno2&~?jmeno) (barva $?barvy3&:(subsetp $?barvy
$?barvy3))))
  =>
  (printout t "Zadany barvam odpovida jediny brouk: " ?jmeno crlf)
  (assert (konec 1))
)

(defrule jeden_brouk_je_prvek
  (declare (auto-focus TRUE) (salience 10))
  (not (konec ?k))
  (barva $?barvy)
  (prvek ?prvky)
  (not (predek ?r))
  (brouk (jmeno ?jmeno) (prvky ?prvky) (barva $?barvy2&:(subsetp $?barvy
$?barvy2)))
  (not(brouk (jmeno ?jmeno2&~?jmeno) (prvky ?prvky) (barva
$?barvy3&:(subsetp $?barvy $?barvy3))))
  =>
  (printout t "Zadany charakteristikam odpovida jediny brouk: " ?jmeno
crlf)
  (assert (konec 1))
)

; ----- Prubezne zjisti neexistence reseni -----

(defrule jeden_brouk_neni_barva
  (declare (auto-focus TRUE) (salience 10))
  (not (konec ?k))
  (barva $?barvy)
  (not (prvek ?p)) (not (predek ?r))
  (not (brouk (barva $?barvy2&:(subsetp $?barvy $?barvy2))))
  =>
  (printout t "Zadane barvy nema zady brouk v databazi." crlf)
  (assert (konec 1))
)

(defrule jeden_brouk_neni_prvek
  (declare (auto-focus TRUE) (salience 10))
  (not (konec ?k))
  (barva $?barvy)
  (prvek ?prvky)
  (not (predek ?r))
  (not (brouk (jmeno ?jmeno) (prvky ?prvky) (barva $?barvy2&:(subsetp
$?barvy $?barvy2))))
  =>
  (printout t "Zadany charakteristikam neodpovida zadny brouk v databazi."
crlf)
  (assert (konec 1))
)

```

```

; ----- Vypsani vysledku hledani -----

(defrule brouci_jsou
  (declare (auto-focus TRUE) (salience 10))
  (not (konec ?k))
  (barva $?barvy)
  (prvek ?prvky)
  (predek ?predky)
  (brouk (jmeno ?jmeno) (predek ?predek) (prvky ?prvky) (barva
  $?barvy2&:(subsetp $?barvy $?barvy2)))
  =>
  (printout t "Hledany brouk by mohl byt: " ?jmeno crlf)
)

(defrule brouci_nejsou
  (declare (auto-focus TRUE) (salience 10))
  (not (konec ?k))
  (barva $?barvy)
  (prvek ?prvky)
  (predek ?predky)
  (not(brouk (jmeno ?jmeno) (predek ?predek) (prvky ?prvky) (barva
  $?barvy2&:(subsetp $?barvy $?barvy2))))
  =>
  (printout t "Brouk se zadanymi vlastnostmi neni v databazi." crlf)
)

(defrule po_vypsani_vysledku
  (declare (auto-focus TRUE) (salience 0))
  (not (konec ?k))
  (barva $?barvy)
  (prvek ?prvky)
  (predek ?predky)
  =>
  (assert (konec 1))
)

```

```

; ----- Dotaz na pokracovani -----

(defrule ma_se_pokracovat
  ?k <- (konec 1)
  =>
  (bind ?y 2)
  (while (or (not(integerp ?y)) (> ?y 1) (< ?y 0)) do
    (printout t crlf "Prejete si identifikovat dalsiho brouka (1 = ano, 0 =
ne - konec)?" crlf)
    (bind ?y(read))
  )
  (retract ?k)
  (assert (konec (+ ?y 2)))
)

(defrule koncime
  (konec 2)
  =>
  (printout t "KONEC" crlf crlf)
)

(defrule pokracujem_a
  (declare (auto-focus TRUE) (salience 10))
  (konec 3)
  ?b <- (barva $?barvy)
  =>
  (retract ?b)
)

(defrule pokracujem_b
  (declare (auto-focus TRUE) (salience 10))
  (konec 3)
  ?p <- (prvek ?prvky)
  =>
  (retract ?p)
)

(defrule pokracujem_c
  (declare (auto-focus TRUE) (salience 10))
  (konec 3)
  ?r <- (predek ?predky)
  =>
  (retract ?r)
)

(defrule pokracujem_d
  ?k <- (konec 3)
  =>
  (retract ?k)
  (printout t crlf)
)

```